

EXAM FUNCTIONAL PROGRAMMING

Tuesday the 30th of September 2014, 11.00 h. - 13.00 h.

Name:

Student number:

Before you begin: Do not forget to write down your name and student number above. If necessary, explain your answers (in English or Dutch). For multiple choice questions, clearly circle what you think is the (one and only) best answer. Use the empty boxes under the other questions to write your answer and explanations in; if you run out of space, you can use the empty sixth page of the exam. Use the empty paper provided with this exam only as scratch paper (kladpapier). At the end of the exam, only hand in the filled-in exam paper. Answers will not only be judged for correctness, but also for clarity and conciseness. A total of one hundred points can be obtained; divide by 10 to obtain your grade. Good luck!

In any of your answers below you may (but do not have to) use the following well-known Haskell functions/operators: *id*, *concat*, *foldr* (and variants), *map*, *filter*, *const*, *flip*, *fst*, *snd*, *not*, *(.)*, *elem*, *take*, *drop*, *takeWhile*, *dropWhile*, *head*, *tail*, *(++)*, *lookup* and all members of the type classes *Eq*, *Num*, *Ord*, *Show* and *Read*.

- (i) Write a function $intersperse :: a \rightarrow [a] \rightarrow [a]$, which places its first argument between the elements of its second argument; i.e. $intersperse 'a' "xyz"$ should return $"xayaz"$. You must use direct recursion.

.../8 There are quite a few possible solutions. The most straightforward one is e.g.:

$$\begin{aligned} intersperse\ a\ (x : y : ys) &= x : a : intersperse\ a\ (y : ys) \\ intersperse\ _ \ xs &= xs \end{aligned}$$

The second case takes care of the empty xs and an xs of length 1.

- (ii) Give an alternative definition of $intersperse$ without direct recursion, using higher-order functions.

.../8 Alternative solutions are:

$$\begin{aligned} intersperse\ a\ xs &= tail\ (foldr\ (\lambda x\ r \rightarrow a : x : r)\ []\ xs) \\ intersperse\ a &= tail\ . foldr\ (\lambda x\ r \rightarrow a : x : r)\ [] \\ intersperse\ a &= tail\ . foldr\ (\lambda x \rightarrow (a:) . (x:))\ [] \\ intersperse\ a &= tail\ . foldr\ ((a:) .) . (:)\ [] \\ intersperse\ a &= tail\ . concat\ . map\ (\lambda x \rightarrow [a, x]) \end{aligned}$$

2. The operator $(.)$ composes two functions. We want to generalise this and implement a function that composes a list of functions *compoR2L* of type $[(a \rightarrow a)] \rightarrow a \rightarrow a$. For example, the call *compoR2L* $[f, g, h]$ v computes the value $f (g (h v))$.

(i) Implement *compoR2L* using *foldr*.

```
.../8
compoR2L :: [a -> a] -> (a -> a)
compoR2L = foldr (.) id
```

(ii) Implement *compoL2R*, that composes functions in the opposite direction. In other words, *compoL2R* $[f, g, h]$ v equals $h (g (f v))$.

```
.../8
compoL2R :: [a -> a] -> (a -> a)
compoL2R = foldl (flip (.)) id
Using (i) and some kind of list-reversal is also okay. Or should I say foldl?
```

(iii) What do you get when you evaluate *compoL2R* $[not, even]$ 3?

```
.../5 A type error since not and even do not have the same type.
```

3. Given is the following definition of a so-called *Trie a*

```
data Trie a = Leaf a | Branch a [(Char, Trie a)]
```

The idea of a *Trie* is that every branch and leaf contains a payload value of type *a*, and that the children of a branch are indexed by a value of type *Char*. An example of a *Trie Int* is the following:

```
ex = Branch 40 [('a', Branch 20 [('a', Leaf 1),
                                ('b', Leaf 2)]),
               ('b', Branch 30 [('a', Leaf 3),
                                ('c', Leaf 4)])
]
```

(i) Write a function *sumIntTrie* :: *Trie Int* -> *Int* that adds all the payloads (of type *Int*) together. For example, *sumIntTrie* (Leaf 3), should return 3, and *sumIntTrie* *ex* should return 100.

.../10

```
sumIntTrie :: Trie Int → Int
sumIntTrie (Leaf v) = v
sumIntTrie (Branch i children) =
  i + sum (map (sumIntTrie . snd) children)
```

- (ii) Write a function `searchTrie :: [Char] → Trie a → Maybe a`, that follows a path down the tree as indicated by the first argument, and just returns the payload of the *branch or leaf* it reaches in this way, and *Nothing* otherwise. For example, `searchTrie [] ex` gives *Just* 40, `searchTrie ['b'] ex` gives *Just* 30, `searchTrie ['b', 'a'] ex` returns *Just* 3, and `searchTrie ['b', 'a', 'h'] ex` returns *Nothing*. Here you may use a function `lookup :: Char → [(Char, b)] → Maybe b` such that `lookup c ps` returns *v* if (c, v) is the first pair in *ps* in which *c* is the first component, and *Nothing* if no such pair exists.

.../10

```
searchTrie :: [Char] → Trie a → Maybe a
searchTrie [] (Leaf v) = Just v
searchTrie [] (Branch v _) = Just v
searchTrie _ (Leaf _) = Nothing
searchTrie (x : xs) (Branch _ chds) =
  case lookup x chds of
    Nothing → Nothing
    (Just child) → searchTrie xs child
```

- (iii) A problem with the definition of *Trie* is that the type system does not forbid values like `Branch 45 [('a', Leaf 33), ('a', Leaf 78)]`. Give a definition for *Trie a* that does not have that problem.

.../5

Use a function: `data Trie a = Leaf a | Branch a (Maybe (Char → Trie a))` but `data Trie a = Leaf a | Branch a (Char → Trie a)` is also okay.

4. .../20 (a), (a), (c), (d)

The following multiple choice questions are each worth 5 points.

(i) Let f be any function of type $Int \rightarrow Int$. Which expression has the same value as the following list comprehension?

$[f\ x \mid x \leftarrow [1..6],\ even\ x]$

- a. $map\ f\ (filter\ even\ [1..6])$
- b. $filter\ even\ (map\ f\ [1..6])$
- c. $f\ (map\ even\ [1..6])$
- d. $filter\ f\ (map\ even\ [1..6])$

(ii) Given is the following unnecessarily complicated function definition:

$f\ g = f\ 0\ g$
where $f\ g\ h \mid g < length\ h = foldr\ (const\ (+1))\ 0\ (h\ !!\ g) + f\ (g + 1)\ h$
 $\mid otherwise = 0$
 $const\ f\ g = f$

Which of the following implementations is equivalent?

- a. $sum\ .\ map\ length$
- b. $foldr\ (+)\ 0\ .\ map\ (const\ 1)$
- c. $foldr\ ((+)\ length)\ 0$
- d. $foldl1\ (+)\ .\ map\ length$

(iii) I Both function application and the \rightarrow in function types associate to the left so that Currying becomes possible.

II Function application has precedence over all operators.

- a. Both I and II are true
- b. Only I is true
- c. Only II is true
- d. Both I and II are false

(iv) What is the type of $map\ .\ foldr$?

- a. $(a \rightarrow a \rightarrow a) \rightarrow [a] \rightarrow [[a] \rightarrow a]$
- b. $(a \rightarrow a \rightarrow a) \rightarrow [b] \rightarrow [b \rightarrow a]$
- c. $(b \rightarrow a \rightarrow a) \rightarrow [b] \rightarrow [[a] \rightarrow a]$
- d. $(b \rightarrow a \rightarrow a) \rightarrow [a] \rightarrow [[b] \rightarrow a]$

5. (i) Explain why the expression $\lambda x \rightarrow (x \text{ '1' }, x \text{ '1' })$ is type incorrect.

.../5 For $x \text{ '1' }$ and $x \text{ '1' }$ to both be type correct, we should derive a polymorphic type for x . But x is lambda-bound, not let-bound, and in that case x can not have a polymorphic type.

- (ii) Determine the type of *map filter*. You should not just write down the type below, but also explain how you arrived at that type (for example, in the way that this is done in the lecture notes of this course).

.../15 See p. 123, sec. 5.10 in the reader. In case of *map filter* we have

$$\begin{aligned} \text{map} &:: (a \rightarrow b) \rightarrow [a] \rightarrow [b] \\ \text{filter} &:: (c \rightarrow \text{Bool}) \rightarrow [c] \rightarrow [c], \end{aligned}$$

where we have chosen fresh type variables in the type of *filter*. The function *map* gets a single argument:

$$\text{map} :: \underbrace{(a \rightarrow b)}_{\text{argumenttype}} \rightarrow \underbrace{[a] \rightarrow [b]}_{\text{resulttype}}.$$

matching the types of the arguments,

$$\underbrace{a \rightarrow b}_{\text{argumenttype of map}} \equiv \underbrace{(c \rightarrow \text{Bool}) \rightarrow [c] \rightarrow [c]}_{\text{type of filter}}$$

$$\Rightarrow \begin{cases} a \equiv c \rightarrow \text{Bool} \\ b \equiv [c] \rightarrow [c], \end{cases}$$

gives:

$$\text{map} :: \underbrace{(a \xrightarrow{(c \rightarrow \text{Bool})} b)}_{\text{argumenttype}} \rightarrow \underbrace{[a] \xrightarrow{[c] \rightarrow [c]} [b]}_{\text{resulttype}}.$$

and thus

$$\text{map} :: \underbrace{((c \rightarrow \text{Bool}) \rightarrow [c] \rightarrow [c])}_{\text{type of filter}} \rightarrow \underbrace{[c \rightarrow \text{Bool}] \rightarrow [[c] \rightarrow [c]]}_{\text{type of map filter}}$$

Finally we get for *map filter*

$$\text{map filter} :: [c \rightarrow \text{Bool}] \rightarrow [[c] \rightarrow [c]],$$

or:

$$\text{map filter} :: [a \rightarrow \text{Bool}] \rightarrow [[a] \rightarrow [a]].$$